*** Settings ***
Documentation  用户登录失败的场景

Resource  ../../resources/user_resource.robot
Resource  ../../resources/action_resource.robot

SUITE SETUP     打开浏览器到首页
SUITE TEARDOWN      关闭浏览器
Test Setup          刷新页面
#Test Teardown       删除网页cookies
Test template       用户登录失败

*** test cases ***              用户名            密码           错误提示信息
使用空的用户名密码登录           ${empty}          123456          信息不完整。
使用用户名和空的密码登录         testuser6        ${empty}         信息不完整。
使用错误的用户名和密码登录        testuser6        123456543         用户名或密码错误
使用特殊字符的用户名和密码       \#%$#$%^&"        "%$#$%^&^%$"      用户名或密码错误
# 如果#在第一行，可以在#前加转译符\


*** keywords ***
用户登录失败
    [Arguments]    ${username}    ${password}       ${error_msg}
    最大化窗口
    点击登录链接
    使用用户名密码进行用户登录      ${username}        ${password}
    用户应该登录失败        ${error_msg}